Mecánica de Medios Continuos. Ingeniero Geólogo (curso 3º). 2006-07

LABORATORIO DE MECÁNICA DE MEDIOS CONTINUOS

Primera Sesión. Preliminares Matemáticos

Miércoles 15 de noviembre de 2006 (8:30 - 9:30)

Javier Rodríguez y José M.ª Goicolea

Breve introducción a Maple

Asignaciones y operadores básicos

```
> a := sqrt(1+x)/2;
```

$$a := \frac{\sqrt{1+x}}{2}$$

```
> a; A;
```

$$\frac{\sqrt{1+x}}{2}$$

$$A$$

```
> b := sqrt(1-x)/2;
```

$$b := \frac{\sqrt{1-x}}{2}$$

```
> x := 2;
```

$$x := 2$$

```
> eval(a);
```

$$\frac{\sqrt{3}}{2}$$

```
> Digits := 20;
```

$$Digits := 20$$

```
> evalf(a);
```

$$0.86602540378443864675$$

```
> eval(b);
```

$$\frac{1}{2} I$$

```
> restart;
> a;
```

$$a$$

Tipos básicos de objetos

Secuencias

```
> 1^2, 2^2, 3^2, 4^2;
```

1, 4, 9, 16

```
> seq(i^2, i=1..4);
```

1, 4, 9, 16

Conjuntos y listas

```
> C := {1, 3, 5};
```

$C := \{1, 3, 5\}$

```
> L1 := [1, 2, 3];
```

$L1 := [1, 2, 3]$

```
> {1, 1, 2, 4}; [1, 1, 2, 4];
```

$\{1, 2, 4\}$

$[1, 1, 2, 4]$

```
> a := L1[2];
```

$a := 2$

```
> L2 := [L1, op(L1), C];
```

$L2 := [[1, 2, 3], 1, 2, 3, \{1, 3, 5\}]$

```
> type(%, list);
```

true

Vectores, matrices y arrays

```
> v := vector([1,4,9]);
```

$v := [1, 4, 9]$

```
> M := matrix([[1, 2, 3], [2, 3, 4], [3, 4, 5]]);
```

$$M := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 3 & 4 \\ 3 & 4 & 5 \end{bmatrix}$$

```
> A := array(2..4, [1, 4, 9]);
```

A := array(2 .. 4, [

(2) = 1

(3) = 4

(4) = 9

])

Álgebra vectorial con el paquete "LinearAlgebra"

```
> with(LinearAlgebra);
```

[*&x, Add, Adjoint, BackwardSubstitute, BandMatrix, Basis, BezoutMatrix, BidiagonalForm, BilinearForm, CharacteristicMatrix, CharacteristicPolynomial, Column, ColumnDimension, ColumnOperation, ColumnSpace, CompanionMatrix, ConditionNumber, ConstantMatrix, ConstantVector, Copy, CreatePermutation, CrossProduct, DeleteColumn, DeleteRow, Determinant, Diagonal, DiagonalMatrix, Dimension, Dimensions, DotProduct, EigenConditionNumbers, Eigenvalues, Eigenvectors, Equal, ForwardSubstitute, FrobeniusForm, GaussianElimination, GenerateEquations, GenerateMatrix, GetResultDataType, GetResultShape,* *GivensRotationMatrix, GramSchmidt, HankelMatrix, HermiteForm, HermitianTranspose, HessenbergForm, HilbertMatrix, HouseholderMatrix, IdentityMatrix, IntersectionBasis, IsDefinite, IsOrthogonal, IsSimilar, IsUnitary, JordanBlockMatrix, JordanForm, LA_Main, LUDecomposition, LeastSquares, LinearSolve, Map, Map2, MatrixAdd, MatrixExponential, MatrixFunction, MatrixInverse, MatrixMatrixMultiply, MatrixNorm, MatrixPower, MatrixScalarMultiply, MatrixVectorMultiply, MinimalPolynomial, Minor, Modular, Multiply, NoUserValue, Norm, Normalize, NullSpace, OuterProductMatrix, Permanent, Pivot, PopovForm, QRDecomposition, RandomMatrix, RandomVector, Rank, RationalCanonicalForm, ReducedRowEchelonForm, Row, RowDimension, RowOperation, RowSpace, ScalarMatrix, ScalarMultiply, ScalarVector, SchurForm, SingularValues, SmithForm, SubMatrix, SubVector, SumBasis, SylvesterMatrix, ToeplitzMatrix, Trace, Transpose, TridiagonalForm, UnitVector, VandermondeMatrix, VectorAdd, VectorAngle, VectorMatrixMultiply, VectorNorm, VectorScalarMultiply, ZeroMatrix, ZeroVector, Zip*]

```
> M1 := Matrix([[1, 2, 3], [2, 3, 5], [3, 4, 5]]);
```

$$M1 := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 3 & 5 \\ 3 & 4 & 5 \end{bmatrix}$$

```
> M2 := <<1, 2, 3>|<2, 3, 5>|<3, 4, 5>>;
```

$$M2 := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 3 & 4 \\ 3 & 5 & 5 \end{bmatrix}$$

```
> type(%, Matrix);
```

$$true$$

```
> v := <1, 4, 9>;
```

$$v := \begin{bmatrix} 1 \\ 4 \\ 9 \end{bmatrix}$$

```
> M := M1 + M2;
```

$$M := \begin{bmatrix} 2 & 4 & 6 \\ 4 & 6 & 9 \\ 6 & 9 & 10 \end{bmatrix}$$

```
> Multiply (M, v);
```

$$\begin{bmatrix} 72 \\ 109 \\ 132 \end{bmatrix}$$

```
> Transpose(M);
```

$$\begin{bmatrix} 2 & 4 & 6 \\ 4 & 6 & 9 \\ 6 & 9 & 10 \end{bmatrix}$$

```
> CharacteristicPolynomial(M, lambda);
```

$$\lambda^3 - 18\,\lambda^2 - 41\,\lambda - 14$$

```
> evalf(Eigenvalues(M));
```

$$\begin{bmatrix} 20.07688221 - 0.4\;10^{-9}\,I \\ -1.655725883 - 0.5862177830\;10^{-8}\,I \\ -0.4211563250 + 0.6262177830\;10^{-8}\,I \end{bmatrix}$$

```
> evalf(Eigenvectors(M));
```

$$\begin{bmatrix} 20.07688221 - 0.4\;10^{-9}\,I \\ -1.655725883 - 0.5862177830\;10^{-8}\,I \\ -0.4211563250 + 0.6262177830\;10^{-8}\,I \end{bmatrix},$$

$[0.5051505119 - 0.2209772641\;10^{-10}\,I,\ -0.8287445010 + 0.2742604124\;10^{-8}\,I,$
$5.573593906 - 0.5681785732\;10^{-6}\,I]$
$[0.7828865744 - 0.1503795506\;10^{-9}\,I,\ -0.7425843208 - 0.1291990302\;10^{-8}\,I,$
$-4.873635506 + 0.3526379931\;10^{-6}\,I]$
$[1.\,,\ 1.\,,\ 1.]$

Funciones y subrutinas

```
> f := x -> x^2;
```

$$f := x \rightarrow x^2$$

```
> f(2);
```

$$4$$

```
> g := (x,y) -> x^y;
```

$$g := (x, y) \rightarrow x^y$$

```
> g(2, 3);
```

$$8$$

```
> g := proc (a, b, c, d, e::uneval) local s;
> s := a + b + c +d: e := s^2:
> end proc;
```

g := **proc**(*a*, *b*, *c*, *d*, *e*::*uneval*) **local** *s*; *s* := *a* + *b* + *c* + *d*; *e* := *s*^2 **end proc**

```
> g(1, 2, 3, 4, e):
> e;
```

$$100$$

```
> map (x -> x^2, [1, 2, 3]);
```

$$[1, 4, 9]$$

```
> solve (x^2 + x - 1 = 0);
```

$$-\frac{1}{2}+\frac{\sqrt{5}}{2},\ -\frac{1}{2}-\frac{\sqrt{5}}{2}$$

```
> fsolve(sin(x) - 2*cos(x), x=1);
```

$$1.107148718$$

```
> diff(f(x),x);
```

$$2\,x$$

Control del flujo

```
> s := 0;
```

$$s := 0$$

```
> for i from 1 by 2 to 100 do s := s + i^2 end do:
> s;
```

$$166650$$

```
> s := 0; i := 0;
```

$$s := 0$$

$$i := 0$$

```
> while i<50 do s := s + (2*i + 1)^2; i := i + 1 end do:
> s;
```

$$166650$$

Graficos

```
> plot(sin(x)*exp(-0.2*x),x=0..6*Pi,thickness=3);
```




```
> plot3d(sin(x)*sin(y), x=Pi/2..9/2*Pi, y=Pi/2..9/2*Pi);
```



```
> plot3d((1.3)^x * sin(y), x=-1..2*Pi, y=0..Pi,
  coords=spherical);
  plot3d([1,x,y], x=0..2*Pi, y=0..2*Pi,
  coords=toroidal(10),scaling=constrained);
```





```
>
```

Ejercicio propuesto*

Determine la inversa y el determinante de la matriz <<1, 2, 3>|<4, 5, 6>|<7, 8, 10>>
Ayúdese de la documentación del programa
Pista: Busque información acerca de los comandos "MatrixInverse" y "Determinant"

```
>
```

Problema 1 (Tema 0)

Sea una base ortonormal a derechas {e1, e2, e3}. Se pide:

1) Demostrar que los vectores u = e1 + e2 - e3 y v = e1 - e2 son ortogonales.

```
> restart;
> with(LinearAlgebra):
> u := <1, 1, -1>;
```

$$u := \begin{bmatrix} 1 \\ 1 \\ -1 \end{bmatrix}$$

```
> v := <1, -1, 0>;
```

$$v := \begin{bmatrix} 1 \\ -1 \\ 0 \end{bmatrix}$$

```
> u . v;
```

$$0$$

2) Determinar una nueva base {g1, g2, g3} de forma que g1 y g2 lleven las direcciones de u y v respectivamente, y esta nueva base forme un triedro ortonormal a derechas.

```
> g1 := Normalize(u, Euclidean);
```

$$g1 := \begin{bmatrix} \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \\ -\frac{\sqrt{3}}{3} \end{bmatrix}$$

```
> g2 := Normalize(v, Euclidean);
```

$$g2 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

```
> g3 := CrossProduct(g1, g2);
```

$$g3 := \begin{bmatrix} -\frac{\sqrt{3}\sqrt{2}}{6} \\ -\frac{\sqrt{3}\sqrt{2}}{6} \\ -\frac{\sqrt{3}\sqrt{2}}{3} \end{bmatrix}$$

3) Determinar la matriz de transformación que permite obtener la nueva base, mediante los coeficientes gi = ep Api.

```
> A := <<g1>|<g2>|<g3>>;
```

$$A := \begin{bmatrix} \frac{\sqrt{3}}{3} & \frac{\sqrt{2}}{2} & -\frac{\sqrt{3}\sqrt{2}}{6} \\ \frac{\sqrt{3}}{3} & -\frac{\sqrt{2}}{2} & -\frac{\sqrt{3}\sqrt{2}}{6} \\ -\frac{\sqrt{3}}{3} & 0 & -\frac{\sqrt{3}\sqrt{2}}{3} \end{bmatrix}$$

4) Determinar la relación matricial de cambio de coordenadas, {v}g = [A]^T {v}e.

```
> <vg[1], vg[2], vg[3]> := Transpose(A) . <ve[1], ve[2],
  ve[3]>;
```

$$\langle vg_1, vg_2, vg_3 \rangle := \begin{bmatrix} \frac{1}{3}\sqrt{3}\, ve_1 + \frac{1}{3}\sqrt{3}\, ve_2 - \frac{1}{3}\sqrt{3}\, ve_3 \\ \frac{1}{2}\sqrt{2}\, ve_1 - \frac{1}{2}\sqrt{2}\, ve_2 \\ -\frac{1}{6}\sqrt{3}\sqrt{2}\, ve_1 - \frac{1}{6}\sqrt{3}\sqrt{2}\, ve_2 - \frac{1}{3}\sqrt{3}\sqrt{2}\, ve_3 \end{bmatrix}$$

Problema 2 (Tema 0)

Sean los vectores {u} = (1, 2, 0)^T, {v} = (0, 1, 1)^T, definidos mediante sus coordenadas en una base ortonormal a derechas. Se pide:

1) Obtener su producto escalar y vectorial.

```
> restart:
> with(LinearAlgebra):
> u := <1, 2, 0>;
```

$$u := \begin{bmatrix} 1 \\ 2 \\ 0 \end{bmatrix}$$

```
> v := <0, 1, 1>;
```

$$v := \begin{bmatrix} 0 \\ 1 \\ 1 \end{bmatrix}$$

```
> uv := u . v;
```

$$uv := 2$$

```
> uxv := CrossProduct(u, v);
```

$$uxv := \begin{bmatrix} 2 \\ -1 \\ 1 \end{bmatrix}$$

2) Se realiza un cambio de base consistente en una rotación de +45º alrededor del eje z(=x3); obtener la matriz de cambio de coordenadas [A]^T, así como las nuevas coordenadas de los vectores {u'} y {v'}.

```
> A := <<cos(Pi/4), sin(Pi/4), 0>|<-sin(Pi/4), cos(Pi/4), 0>|<0, 0, 1>>;
```

$$A := \begin{bmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} & 0 \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> AT := Transpose(A);
```

$$AT := \begin{bmatrix} \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \\ -\frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> u´ := AT . u;
```

$$u´ := \begin{bmatrix} \frac{3\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

```
> v´ := AT . v;
```

$$v´ := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \\ 1 \end{bmatrix}$$

3) Comprobar que el producto escalar calculado con las nuevas componentes se conserva.

```
> uv - u´ . v´;
```

$$0$$

4) Comprobar que las coordenadas del producto vectorial en la nueva base correponden a las antiguas aplicando [A]^T.

```
> CrossProduct(u´, v´) - AT . uxv;
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Problema 3 (Tema 0)*

Se define un cambio de coordenadas mediante una rotación de ángulo \theta alrededor del eje x. Se pide:

1) Obtener la matriz de cambio de coordenadas [A] = [Aij] mediante la aplicación de la fórmula Aij = ei . e'j.

```
> restart:
> with(LinearAlgebra):
> e[1] := <1, 0, 0>;
```

$$e_1 := \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$$

```
> e[2] := <0, 1, 0>;
```

$$e_2 := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$$

```
> e[3] := <0, 0, 1>;
```

$$e_3 := \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$$

```
> e´[1] := <cos(theta), sin(theta), 0>;
```

$$e´_1 := \begin{bmatrix} \cos(\theta) \\ \sin(\theta) \\ 0 \end{bmatrix}$$

```
> e´[2] := <-sin(theta), cos(theta), 0>;
```

$$e´_2 := \begin{bmatrix} -\sin(\theta) \\ \cos(\theta) \\ 0 \end{bmatrix}$$

```
> e´[3] := <0, 0, 1>;
```

$$e´_3 := \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$$

```
> A := Matrix(3, 3);
```

$$A := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> for i from 1 to 3 do
> for j from 1 to 3 do
> A[i, j] := e[i] . e´[j]
> od
> od:
> A;
```

$$\begin{bmatrix} \cos(\theta) & -\sin(\theta) & 0 \\ \sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

2) Comprobar que la matriz así obtenida es ortogonal.

```
> simplify(A . Transpose(A));
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

3) Demostrar que [A]^n corresponde a una rotación de ángulo n \theta.

```
> B := subs(theta=n*theta, A) - subs(theta=(n-1)*theta,
  A).A;
```

$B :=$

$[\cos(n\,\theta) - \cos((n-1)\,\theta)\cos(\theta) + \sin((n-1)\,\theta)\sin(\theta),$
$-\sin(n\,\theta) + \sin((n-1)\,\theta)\cos(\theta) + \cos((n-1)\,\theta)\sin(\theta), 0]$
$[\sin(n\,\theta) - \cos((n-1)\,\theta)\sin(\theta) - \sin((n-1)\,\theta)\cos(\theta),$
$\cos(n\,\theta) - \cos((n-1)\,\theta)\cos(\theta) + \sin((n-1)\,\theta)\sin(\theta), 0]$
$[0, 0, 0]$

```
> Map(x->simplify(expand(x)), B);
```

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
>
>
```